

# Excel Date & Time

---

EVERYTHING YOU NEED TO KNOW ABOUT WORKING  
WITH DATES AND TIME IN EXCEL

Mynda Treacy

MICROSOFT EXCEL MVP AND CO-FOUNDER MY ONLINE TRAINING HUB

## CONTENTS

Introduction .....	4
Regional Settings .....	4
Excel Date & Time 101 .....	5
In a nutshell .....	5
Dates .....	5
Time .....	6
Date & Time Together .....	6
Good to know .....	7
Entering Dates & Times in Excel .....	8
Entering Dates .....	8
Entering Dates with Two Digit Years .....	9
Entering Time .....	9
Entering Dates & Time Together .....	10
Simple Date & Time Math .....	11
Adding/Subtracting Days from Dates .....	11
Subtracting Dates from one another .....	11
Adding Times to one another .....	12
Subtracting Time from Times .....	12
Subtracting Times from one another .....	12
Date & Time Shortcuts .....	13
Formatting Dates & Time .....	14
Custom Formats .....	15
Custom Date Formatting Characters .....	17
Custom Date Formatting Examples .....	17
Custom Time Formatting Characters .....	18
Custom Time Formatting Examples .....	19
Identifying if Dates are Entered as Text .....	20
Fixing Text Dates .....	22
Converting Date Serial Numbers to Text .....	26
Extract Month Number from Text Month Name .....	27
Extract Date/Time Elements .....	29
Common Date Calculations .....	29
Calculate Age or Years of Service .....	29

Difference between Dates .....	31
Finding Dates .....	32
Difference between Times (same day).....	32
Difference between Times (different days).....	33
Rounding Time and Time Intervals (using CEILING & FLOOR).....	33
Rounding Time and Time Intervals (using MROUND).....	34
Convert Time to Decimals .....	34
Convert Minutes to Hours, Hours to Days etc. in Decimals.....	34
Excel Date and Time Functions .....	35
DATE Function.....	38
TIME Function.....	40
DATEVALUE Function .....	41
TIMEVALUE Function .....	42
NOW Function.....	43
TODAY Function .....	44
HOUR Function .....	45
MINUTE Function .....	46
SECOND Function .....	47
DAY Function .....	48
MONTH Function.....	49
YEAR Function .....	50
WEEKNUM Function .....	51
ISOWEEKNUM Function .....	53
WEEKDAY Function .....	54
EDATE Function.....	55
EOMONTH Function .....	56
WORKDAY Function.....	57
WORKDAY.INTL Function.....	58
DAYS Function.....	60
DAYS360 Function .....	61
NETWORKDAYS Function .....	62
NETWORKDAYS.INTL Function.....	63
YEARFRAC Function .....	65
DATEDIF Function (use with caution).....	66

DATEDIF Alternatives.....	68
More resources.....	69
About.....	71
Thanks .....	71

## INTRODUCTION


The objective of this book is to teach you how Excel handles date and time and provide you with all the tools you will need.

It's designed to be read in conjunction with the accompanying Excel file, which you can [download here](#).

## REGIONAL SETTINGS

When reading this eBook keep in mind that my regional settings format dates as dd/mm/yyyy and so the screenshots throughout this book are in this format. However, if you open the accompanying Excel file you may see some dates have switched to match your regional settings, which may be different to mine e.g. mm/dd/yyyy.

Dates and times with a format that begins with an asterisk (\*) automatically update based on your PC's regional settings. You can see an example in the Format Cells dialog box below:




## EXCEL DATE & TIME 101

### IN A NUTSHELL

Excel stores dates and time as a number known as the date serial number, or date-time serial number.

When you look at a date in Excel it's actually a regular number that has been formatted to look like a date. If you change the cell format to 'General' you'll see the underlying date serial number.

The integer portion of the date serial number represents the day, and the decimal portion is the time. Dates start from 1st January 1900 i.e. 1/1/1900 has a date serial number of 1.

Formatted date & time	1/01/2012 10:00 AM
Underlying serial number	40909.416666667
	



**Caution!** Excel dates after 28th February 1900 are actually one day out. Excel behaves as though the date 29th February 1900 existed, which it didn't.

Microsoft intentionally included this bug in Excel so that it would remain compatible with the spreadsheet program that had the majority market share at the time; Lotus 1-2-3.

Lotus 1-2-3 was incorrectly programmed as though 1900 was a leap year. This isn't a problem as long as all your dates are later than 1st March 1900.

### DATES

Excel gives each date a numeric value starting at 1<sup>st</sup> January 1900. 1<sup>st</sup> January 1900 has a numeric value of 1, the 2<sup>nd</sup> January 1900 has a numeric value of 2 and so on. These are called 'date serial numbers', and they enable us to do math calculations and use dates in formulas.

Date	Date Serial Number
1/01/1900	1
2/01/1900	2
1/02/1900	32
30/07/1960	22127
1/01/2017	42736
1/04/2018	43191

The Date Serial Number column displays the Date column values in their date serial number equivalent.

e.g. 1/1/2017 has a date serial number of 42736. i.e. 1<sup>st</sup> January 2017 is 42,736 days since 31<sup>st</sup> December 1899.

**Tip:** format the date serial number column as a Date and you'll see they look the same as the Date column values.

## TIME

Times also use a serial number format and are represented as decimal fractions.

**Hours:** since 24 hours = 1 day, we can infer that 24 hours has a time serial number of 1, which can be formatted as time to display 24:00 or 12:00 AM or 0:00. Whereas 12 hours or the time 12:00 has a value of 0.50 because it is half of 24 hours or half of a day, and 1 hour is 0.041666' because it's 1/24 of a day.

**Minutes:** since 1 hour is 1/24 of a day, and 1 minute is 1/60 of an hour, we can also say that 1 minute is 1/1440 of a day, or its time serial number is 0.00069444'

**Seconds:** since a second is 1/60 of a minute, which is 1/60 of an hour, which is 1/24 of a day, we can also say one second is 1/86400 of a day or in time serial number form it's 0.0000115740740740741...

Time (text)	Time Serial Numbers Formatted h:mm:ss	Time Serial Numbers formatted General
1 day or 24 hours	0:00:00	1
1 hour	1:00:00	0.041666667
5 hours	5:00:00	0.208333333
1 minute	0:01:00	0.000694444
30 seconds	0:00:30	0.000347222
1 hr 30 min 45 seconds	1:30:45	0.063020833

## DATE & TIME TOGETHER

Now that we know how dates and times are stored we can put them together - dddd.ttttt

For example, the date and time of 1<sup>st</sup> January 2012 10:00:00 AM has a date-time serial value of 40909.416666667

40909 being the serial value representing the date 1<sup>st</sup> January 2012, and .416666667 being the decimal value for the time 10:00 AM and 00 seconds.

More examples below.

Date-Time	Serial Numbers Formatted as dd/mm/yyyy h:mm:ss	Serial Numbers formatted General
1st January 1900 10:00 AM	01/01/1900 10:00:00	1.41667
3rd March 1960 1:00 PM	03/03/1960 13:00:00	21978.54167
30th April 2017 11:00 PM	30/04/2017 23:00:00	42855.95833

## GOOD TO KNOW

- Dates prior to 1st January 1900 are not recognised in Excel.
- A negative date will display in the cell as #####
- Times stored without a date effectively inherit the date 0 Jan 1900 i.e. the month is Jan and the year 1900 and the day is zero. Remember, there are no dates prior to 1/1/1900 from Excel's perspective. This means that times stored without a date e.g. 0.50 for 12:00 PM is the equivalent of 0 Jan 1900 12:00 PM.

This is important because if you try to take 14 hours from 12 hours (without a date) you'll get the dreaded ##### display in the cell, because negative dates and times cannot be displayed. We'll cover workarounds for this later, but for now keep in mind that math on dates and time that result in negative date-time serial numbers cannot be formatted as a date.

- Excel actually has two date modes. The other mode is called 1904 Date System and is used for compatibility with Excel 2008 for Mac and earlier Mac versions. You can change the date system in the Advanced Options.

In the 1904 date system dates are calculated using 1st January 1904 as the starting point. The difference between the two date systems is 1,462 days. This means that the serial number of a date in the 1900 date system is always 1,462 days greater than the serial number of the same date in the 1904 date system. 1,462 days is equal to four years and one day (including one leap day).

**Caution;** the date setting you choose applies to all dates within the workbook. You can't mix and match modes and you shouldn't reference workbooks that use a different date system in formulas.

Bottom line; don't use the 1904 date system unless absolutely necessary! [Click here for more on date systems in Excel.](#)

- Excel applies date number formats based on your system region settings. For example, my system is set to display dates in dd/mm/yyyy format, but if you're in the U.S. your system is likely to format them as mm/dd/yyyy. Excel will automatically convert the format of date serial numbers to suit your system settings as long as it's one of the default date formats and not a custom number format.



## ENTERING DATES & TIMES IN EXCEL

### ENTERING DATES

You can type in various configurations of a date and Excel will automatically recognise it as a date and upon pressing ENTER it will convert it to a date serial number and apply a date format on the cell.

For example, try typing (or even copy and paste) the following dates into an empty cell:


	Entered	Excel Returns	Date Serial Number
	1-1-2009	1/01/2009	39814
	1-1-09	1/01/2009	39814
	1/1/2009	1/01/2009	39814
	1/1/09	1/01/2009	39814
	1-Jan-09	1-Jan-09	39814
	1-Jan 09	1-Jan-09	39814
	1-Jan-2009	1-Jan-09	39814
	1 Jan 09	1-Jan-09	39814
	1/1	1-Jan-17	42736

You can see that entering numbers that look like dates which are separated by a forward slash or hyphen will be recognised as a date. Even typing in a date with the month name gets converted to a date.

However, dates separated with a period like this 1.1.2009, or spaces with between numbers like this 01 01 2009, will end up as text, not a date. Gotta have some limits!

**Tip:** Dates that display ##### in a cell usually indicate that the column is simply not wide enough to display it.

However, if you make the cell really wide and it still displays ##### then this indicates that the date is a negative value and Excel can't display negative dates



## ENTERING DATES WITH TWO DIGIT YEARS

When you enter a date with two digits for the year e.g. 1/1/09, Excel has to decide if you mean 2009 or 1909.

It goes by the rule that dates with years 29 or before, are treated as 20xx and dates with the year 30 or older are treated as 19xx. See examples below.

Entered	Excel Returns	Date Serial Number
1/1/29	1/01/2029	47119
1/1/30	1/01/1930	10959
1/1/40	1/01/1940	14611
1/1/15	1/01/2015	42005
1/1/01	1/01/2001	36892

**Tip:** You can enter the day and month portions of a date and Excel will insert the year based on your computer's clock. Nice to know for data entry.

## ENTERING TIME

When you enter time you must follow a strict format of at least h:mm. i.e. the hour and minutes are separated by a colon with no spaces either side. Entering the h:mm components will result in a time formatted in military time e.g. 2:00 PM is 14:00 in military time.

If you enter a time that includes a seconds component e.g. 3:15:40, Excel will automatically format the cell in h:mm:ss.

If you want the time to be formatted with AM/PM you can simply enter a space after the time and then type AM or PM, or apply the number format to the cell later. Here are some examples:

Entered	Excel Returns	Time Serial Number
9:00	9:00	0.375
1:00	1:00	0.041666667
13:00	13:00	0.541666667
12:30:45	12:30:45	0.521354167
1:00 PM	1:00 PM	0.541666667
11:15:30 AM	11:15:30 AM	0.469097222
16 : 45	0.697916667	0.697916667
4 : 45 PM	0.697916667	0.697916667

## ENTERING DATES & TIME TOGETHER

Now that we know how to enter dates and time separately we can put them together to enter a date and time in the same cell.

You can even enter time then date and Excel will fix the order for you.

Entered	Excel Returns	Time	Serial Number
01/01/2018 1:00:00 AM	1/01/2018 1:00		43101.04167
1/1/2017 2:00:30 PM	1/01/2017 14:00		42736.58368
2:00 PM 1/1/17	1/01/2017 14:00		42736.58333

You'll find that even if you enter AM/PM, that Excel will convert it to military time by default. You can override this with a custom number format. More on that later.

## SIMPLE DATE & TIME MATH

Now that we understand that Excel stores dates and time as serial numbers, you'll see how logical it is to perform math operations on these values. We'll look at some simple examples here and tackle the more complex scenarios later when we look at Date and Time Functions.

### ADDING/SUBTRACTING DAYS FROM DATES

	B	C	D	E	F
9	<b>Date</b>	<b>+/- Days</b>	<b>Formula</b>	<b>Result</b>	<b>Date Serial Number</b>
10	1/01/2018	10	=B10+C10	11/01/2018	43111
11	20/02/2018	-5	=B11+C11	15/02/2018	43146
12	1/01/2018	365	=B12+C12	1/01/2019	43466
13	1/01/2018	30	=B13+C13	31/01/2018	43131

**Tip:** you can also add/subtract the days directly in the formula e.g. =B10+10 or =B11-5 Although, it's better to place the values you're adjusting by in their own cell or a named range.

### SUBTRACTING DATES FROM ONE ANOTHER

	B	C	D	E
17	<b>Start Date</b>	<b>End Date</b>	<b>Formula</b>	<b>Result (Days)</b>
18	1/01/2018	30/01/2018	=C18-B18	29
19	1/01/2018	3/01/2018	=C19-B19	2
20	1/01/2018	31/12/2018	=C20-B20	364
21	1/01/2018	1/01/2016	=C21-B21	-731

**Tip:** format the cell to General or Number to see the number of days between two dates.

**Note:** the 'result' is exclusive of the start day i.e. it assumes the start day is at the end of that day.

## ADDING TIMES TO ONE ANOTHER

The time being added is input as a time serial number. Notice there are no negative times in the table below. Remember we can't display negative times. Instead we need to use the math operator to tell Excel to subtract time. See examples below.

	B	C	D	E	F
25	<b>Start Time</b>	<b>Add Time</b>	<b>Formula</b>	<b>Result</b>	<b>Time Serial Number</b>
26	9:00:00 AM	8:00:00	=B26+C26	5:00:00 PM	0.708333333
27	6:30:00 AM	12:00:00	=B27+C27	6:30:00 PM	0.770833333
28	10:00:00 PM	4:00:00	=B28+C28	2:00:00 AM	1.083333333
29	10:00:00 PM	4:00:00	=B29+C29	26:00:00	1.083333333

**Note:** Times that roll over to the next day result in a time-date serial number  $\geq 1$ . Cell E28 actually contains a time-serial number of 1.08333', but since the cell is formatted to display time formatted as h:mm:ss, only the time portion is visible.

If you want to show the cumulative time (like cell E29) then you need to surround the 'h' part of the time format in square brackets like so: **[h]:mm:ss**

## SUBTRACTING TIME FROM TIMES

Notice the last result in the table below shows #####, this is because it results in a negative time and Excel can't display that, but notice it can return a negative time serial number. More on how to solve this later.

	B	C	D	E	F
39	<b>Time</b>	<b>h:mm:ss</b>	<b>Formula</b>	<b>Result</b>	<b>Time Serial Number</b>
40	9:00:00 AM	5:00:00	=B40-C40	4:00:00 AM	0.166666667
41	1:00:00 PM	0:30:00	=B41-C41	12:30:00 PM	0.520833333
42	11:00:00 PM	3:00:00	=B42-C42	8:00:00 PM	0.833333333
43	2:00:00 AM	4:00:00	=B43-C43	#####	-0.083333333

## SUBTRACTING TIMES FROM ONE ANOTHER

Again, here the last result shows ##### because it results in a negative time.

	B	C	D	E	F
47	<b>Start Time</b>	<b>End Time</b>	<b>Formula</b>	<b>Result</b>	<b>Time Serial Number</b>
48	9:00:00 AM	1:00:00 PM	=C48-B48	4:00:00 AM	0.166666667
49	1:00:00 PM	11:00:00 PM	=C49-B49	10:00:00 AM	0.416666667
50	6:30:00 AM	3:45:00 PM	=C50-B50	9:15:00 AM	0.385416667
51	11:00:00 PM	4:00:00 AM	=C51-B51	#####	-0.791666667


## DATE &amp; TIME SHORTCUTS

Action	Shortcut Key
Insert Current Date	Ctrl + ;
Insert Current Time	Ctrl + Shift + ;
Apply Date Format	Ctrl + Shift + #
Apply Time Format	Ctrl + Shift + @

## FORMATTING DATES & TIME

Even though dates and time are actually stored as a regular number known as the date serial number, we can make use of extensive formatting options to display them just the way we want.


We can access some quick formats from the Home tab > in the Number group:



## CUSTOM FORMATS


We can also create our own custom date and time formats to suit our needs. Let's take a look.

1. Select the cell(s) containing the dates you want to format.
2. Press CTRL+1, or right-click > Format Cells to open the Format Cells dialog box.
3. On the Number tab select 'Date' in the Categories list. This brings up a list of default date formats you can select from in the 'Type' list. Likewise for the Time category.





We aren't limited to the defaults though. You can create your own Custom date or time formats in the 'Custom' category. These custom formats are saved for you to re-use in the current file.



## CUSTOM DATE FORMATTING CHARACTERS

Excel recognises the following characters and sets of characters for date formatting.

Character	Explanation	Date	Formatted
d	Displays the day as a number without a leading zero.	3/09/2016	3
dd	Displays the day as a number with a leading zero when appropriate.	3/09/2016	03
ddd	Displays the day as an abbreviation (Sun to Sat).	3/09/2016	Sat
dddd	Displays the day as a full name (Sunday to Saturday).	3/09/2016	Saturday
m	Displays the month as a number without a leading zero.	3/09/2016	9
mm	Displays the month as a number with a leading zero when appropriate.	3/09/2016	09
mmm	Displays the month as an abbreviation (Jan to Dec).	3/09/2016	Sep
mmmm	Displays the month as a full name (January to December).	3/09/2016	September
mmmmm	Displays the month as a single letter (J to D).	3/09/2016	S
yy	Displays the year as a two-digit number.	3/09/2016	16
yyyy	Displays the year as a four-digit number.	3/09/2016	2016

## CUSTOM DATE FORMATTING EXAMPLES

We can bring the characters together to create our own custom formats. Some examples below:

Date	Custom Format	Formatted Result
30/03/2017	dddd dd mmm, yyyy	Thursday 30 Mar, 2017
30/03/2017	mmm-yy	Mar-17
30/03/2017	dd-mm-yy	30-Mar-17
30/03/2017	yyyy-mm	2017-03
30/03/2017	ddd dd	Thu 30
30/03/2017	mmm/yyyy	Mar/2017

Remember; the custom format doesn't alter the underlying date serial number, it is still the same.

## CUSTOM TIME FORMATTING CHARACTERS

Like dates, time also has its own set of custom formatting characters, as listed below:

Character	Explanation
h	Displays the hour as a number without a leading zero.
[h]	Displays elapsed time in hours. If you are working with a formula that returns a time in which the number of hours exceeds 24, use a number format that resembles [h]:mm:ss or [h]:mm
hh	Displays the hour as a number with a leading zero when appropriate. If the format contains AM or PM, the hour is based on the 12-hour clock. Otherwise, the hour is based on the 24-hour clock.
m	Displays the minute as a number without a leading zero.*
[m]	Displays elapsed time in minutes. If you are working with a formula that returns a time in which the number of minutes exceeds 60, use a number format that resembles [mm]:ss.
mm	Displays the minute as a number with a leading zero when appropriate.*
s	Displays the second as a number without a leading zero.
[s]	Displays elapsed time in seconds. If you are working with a formula that returns a time in which the number of seconds exceeds 60, use a number format that resembles [ss].
ss	Displays the second as a number with a leading zero when appropriate. If you want to display fractions of a second, use a number format that resembles h:mm:ss.00.
AM/PM, am/pm, A/P, a/p	Displays the hour using a 12-hour clock. Excel displays AM, am, A, or a for times from midnight until noon and PM, pm, P, or p for times from noon until midnight.

\*Note: The m or mm code must appear immediately after the h or hh code or immediately before the ss code; otherwise, Excel displays the month instead of minutes.

CUSTOM TIME FORMATTING EXAMPLES

Time	Format	Formatted Result
9:35:42	h	9
9:35:42	h:mm	9:35
9:35:42	hh:mm	09:35
9:35:42	h:mm:ss	9:35:42
9:35:42	h:mm:ss.00	9:35:42.00
9:35:42	hh:mm:ss AM/PM	09:35:42 AM
9:35:42	h "hours" mm "minutes" ss "seconds"	9 hours 35 minutes 42 seconds



**Note:** if your PC region settings have the Date & Time formats set to show the Short Time as hh:mm tt or the Long Time as hh:mm:ss tt then this may override any single 'h' formats and display them as 'hh'.

The screenshot above is what I see with my PC region settings for the Short Time as h:mm tt. If you see something different when using a single 'h' format, then it will be down to your PC region settings.

## IDENTIFYING IF DATES ARE ENTERED AS TEXT


Dates incorrectly entered or imported as text are a showstopper. We need our dates entered as date serial numbers to be able to leverage the Date Functions, use them in math and lookups, or even use them in PivotTables.

It can be tricky to tell if a date is text, because unlike regular values, a date will often appear the same in the formula bar as it does on the face of the cell. Don't worry, I've got several never fail, dead easy ways to tell if dates are 'dates' or 'text'.

	B	C	D	E	F	G
10	Text Dates dd/mm/yyyy	<p>At first glance the first and second set of dates look identical. Even comparing them by looking in the formula bar doesn't give anything away.</p> 				
11	1/01/2017					
12	2/01/2017					
13	3/01/2017					
14	4/01/2017					
15	5/01/2017	<p>And both sets are formatted with the 'Date' number format:</p> 				
16						
17	Date Serial Numbers					
18	1/01/2017					
19	2/01/2017					
20	3/01/2017					
21	4/01/2017					
22	5/01/2017					
23						


### Quick Test 1:

Only one set will display a sum in the status bar (and average, min and max if you have them set to display):



That's right, the dates that are correctly entered as date serial numbers will display the SUM result in the status bar when you select **more than one cell** containing a date. Whereas text will only display the count.

So, a quick test to check if dates are text is to select more than one cell containing your dates, then glance down to the status bar to see if you get a sum result. The status bar is in the bottom right of the Excel window:



## Quick Test 2:

Another quick test is to change the cell format to General.

If your date format displays the date serial number (as shown in the second list of dates to the right) you're good to go, but if it still displays a date (as shown in the first list of dates on the right) then you've got yourself some text dates that need fixing.

It's not as simple as just changing the cell format to 'Date'. Those text dates are stubborn. I show you how to fix them in the next chapter.

Text Dates  
dd/mm/yyyy

1/01/2017

2/01/2017

3/01/2017

4/01/2017

5/01/2017

Date Serial  
Numbers


42736

42737

42738

42739

42740



The screenshot shows the 'Number' format dropdown menu in Excel. The 'General' option is selected and highlighted with an orange border. Other options include '\$', '%', and decimal formatting options.

## Quick Test 3:

Widen the column; any dates that are text will align to the left, assuming no other alignment formatting has been applied to the cells.

	A
1	<b>Text Dates dd/mm/yyyy</b>
2	1/1/17
3	2/1/17
4	3/1/17
5	4/1/17
6	5/1/17
7	
8	<b>Date Serial Numbers</b>
9	1/01/2017
10	2/01/2017
11	3/01/2017
12	4/01/2017
13	5/01/2017

## FIXING TEXT DATES

### Option 1: Use Built in Error Checking to Fix Dates


More recent versions of Excel are good at detecting dates incorrectly entered as text and tagging them with a warning notification. Hovering your mouse over the warning tells you the date is entered as text:



Clicking on the warning reveals a list of options for fixing the date, or ignoring the error, as you can see in the image on the right:

**Tip:** if you have a lot of dates to fix, or a lot of formulas referencing these dates then converting them to date serial numbers with this method could be slow. Option 2 may be quicker.

In cases where the warning doesn't display, as well as in earlier versions of Excel, we need to resort to other methods of fixing dates formatted as text. Let's take a look.




## Option 2: Paste Special

1. In an empty cell enter 1 and copy the cell to the clipboard
2. Select the cells containing the text dates you want to convert to date serial numbers.
3. Home tab > Paste > 4. Paste Special

**Tip:** Paste Special > Values shortcut keys:

ALT > E > S > V > Enter or

CTRL+ALT+V > V > Enter



In the Paste Special dialog box select:

- Values
- Multiply (or Divide)


**Alternate approach:** You can also use Paste Special with 'Add'.

Simply copy an empty cell > Select your date cells > Paste Special :

Values & Add!

No need to go back and delete the '1'.

These methods will return a list of date serial numbers which you can then apply a date number format to.



**Tip:** on long lists with lots of formulas referencing your dates this method is often quicker than the error checking method in option 1.




## Option 3: Text to Columns

If your text dates aren't formatted based on the standard structure for your region e.g. dd/mm/yyyy or mm/dd/yyyy, then a great tool to use is Text to Columns, because this allows you to specify the order of the date characters ensuring that they are converted to date serial numbers correctly.

1. Select the cells containing your dates
2. Data tab
3. Text to Columns
4. Delimited
5. Next


In step 2 of the wizard simply click Next.



In step 3 of the wizard you can select the order of the date data from the drop down.

**Note:** this is the format of the text you're converting, not your final desired format.

Your final desired format can be applied with custom number formatting.



## Option 4: DATEVALUE Formula

The DATEVALUE function takes a date text string and converts it to a date serial number, as you can see in this example:

	A	B	C
		Text Dates dd/mm/yyyy	DATEVALUE Formulas
1		1/1/17	42736
2		2/1/17	42737
3		3/1/17	42738

**Tip:** You'll probably want to convert those DATEVALUE formulas in column C to values with Paste Special > Values and get rid of the text dates in column B. No need to have them cluttering up your spreadsheet.

## CONVERTING DATE SERIAL NUMBERS TO TEXT

If you want to use a date in a chart label, or you want to concatenate a date with some other text, then you'll want to convert the date serial number to text first to ensure it displays correctly.

We can use the TEXT function for this. The second argument of the text function allows us to specify the date format we want applied.

<b>Syntax:</b>	=TEXT(value, format_text)
<b>value</b>	This is typically a reference to a cell containing the date serial number, but could also be a DATE formula.
<b>format_text</b>	This is the date number format you want to apply (surrounded by double quotes). Tip: Use formatting principles described in the Formatting section.

**Tip:** You can use the date number formats to construct the date's appearance in many different ways. You can also concatenate additional text using the & symbol, as shown in C19 and C20.

	B	C	D
15	Date Serial Numbers	TEXT Formula Result	TEXT Formula in column C
16	1/01/2017	1 Jan 17	=TEXT(B16,"d mmm yy")
17	2/01/2017	Mon 2 Jan 17	=TEXT(B17,"ddd d mmm yy")
18	3/01/2017	03- 01-2017	=TEXT(B18,"dd- mm-yyyy")
19	4/01/2017	Date: 4 Jan 17	="Date: "&TEXT(B19,"d mmm yy")
20	5/01/2017	Due: Jan 05, 17	="Due: "&TEXT(B20,"mmm dd, yy")
21	1/01/2016	1st January, 2016	=DAY(B21)&IF(OR(DAY(B21)={1,2,3,21,22,23,31}), CHOOSE(1*RIGHT(DAY(B21),1),"st ","nd ","rd "), "th")& TEXT(B21,"mmmm, yyyy")
22	2/01/2017	Mon, 2 Jan 2017	=CHOOSE(WEEKDAY(B17), "Sun","Mon","Tue","Wed","Thu","Fri","Sat")&, "&DAY(B17)&" "&CHOOSE(MONTH(B17),"Jan","Feb","Mar","Apr","May", "Jun","Jul","Aug","Sep","Nov","Dec")&" "&YEAR(B17)

**Tip:** The last example on row 22 shown above will always return a date formatted as text in English, irrespective of the PC's regional language settings. Change the values in CHOOSE to return your preferred language.


**Note:** the TEXT function converts the dates to text, obviously! But bear in mind that this means you can't use the values in column C in any math formulas, or many of the Date functions.

## EXTRACT MONTH NUMBER FROM TEXT MONTH NAME

Often, we'll have a list of month names that we want to convert to a date, or even just extract the month number.

	B	C	D
6	Text Month Name	Convert to Month No.	Formula in column C
7	January	1	=MONTH(B7&1)
8	February	2	=MONTH(B8&1)
9	March	3	=MONTH(B9&1)
10	April	4	=MONTH(B10&1)
11	May	5	=MONTH(B11&1)
12	June	6	=MONTH(B12&1)
13	July	7	=MONTH(B13&1)
14	August	8	=MONTH(B14&1)
15	September	9	=MONTH(B15&1)
16	October	10	=MONTH(B16&1)
17	November	11	=MONTH(B17&1)
18	December	12	=MONTH(B18&1)

**How it works:** By concatenating a 1 to the text month name in column B we provide the MONTH function with a text date it can recognise/use. We can see how the formula in cell C7 evaluates in the Evaluate Formula dialog box below:



	B	C	D	E
25	Text Month Name	Convert to Date	Formatted Date	Formula in column C
26	January	42736	1/01/2017	=DATE(YEAR(TODAY()),MONTH(B26&1),1)
27	February	42767	1/02/2017	=DATE(YEAR(TODAY()),MONTH(B27&1),1)
28	March	42795	1/03/2017	=DATE(YEAR(TODAY()),MONTH(B28&1),1)
29	April	42826	1/04/2017	=DATE(YEAR(TODAY()),MONTH(B29&1),1)
30	May	42856	1/05/2017	=DATE(YEAR(TODAY()),MONTH(B30&1),1)
31	June	42887	1/06/2017	=DATE(YEAR(TODAY()),MONTH(B31&1),1)
32	July	42917	1/07/2017	=DATE(YEAR(TODAY()),MONTH(B32&1),1)
33	August	42948	1/08/2017	=DATE(YEAR(TODAY()),MONTH(B33&1),1)
34	September	42979	1/09/2017	=DATE(YEAR(TODAY()),MONTH(B34&1),1)
35	October	43009	1/10/2017	=DATE(YEAR(TODAY()),MONTH(B35&1),1)
36	November	43040	1/11/2017	=DATE(YEAR(TODAY()),MONTH(B36&1),1)
37	December	43070	1/12/2017	=DATE(YEAR(TODAY()),MONTH(B37&1),1)

**Note:** The formula in column C converts the month to the first day of each month in the current year.

**Tip:** it also works with abbreviated dates!

	B	C	D	E
47	Text Month Name	Convert to Date	Formatted Date	Formula in column C
48	Jan	42736	1/01/2017	=DATE(YEAR(TODAY()),MONTH(B48&1),1)
49	Feb	42767	1/02/2017	=DATE(YEAR(TODAY()),MONTH(B49&1),1)
50	Mar	42795	1/03/2017	=DATE(YEAR(TODAY()),MONTH(B50&1),1)
51	Apr	42826	1/04/2017	=DATE(YEAR(TODAY()),MONTH(B51&1),1)
52	May	42856	1/05/2017	=DATE(YEAR(TODAY()),MONTH(B52&1),1)

You can replace the 'YEAR(TODAY())' part of the formula with a different year, if you prefer.

\*Special thanks to Sunny Kow for reminding me about this tip.

## EXTRACT DATE/TIME ELEMENTS

We often need to extract the date or time portion from a date-time serial number for use in other functions. We can use the INT function to extract the date portion, and the MOD function to extract the time portion. See examples below.

	B	C	D
9	Date-Time Serial Number	Extract the Date	Formula
10	01/01/2017 1:00:00 AM	1/01/2017	=INT(B10)
11	01/02/2017 2:00:30 PM	1/02/2017	=INT(B11)
12	03/03/2017 11:30:00 AM	3/03/2017	=INT(B12)
13	04/01/2017 12:00:00 AM	4/01/2017	=INT(B13)
14	05/10/2017 9:00:00 AM	5/10/2017	=INT(B14)
15			
16			
17	Date-Time Serial Number	Extract the Time	Formula
18	01/01/2017 1:00:00 AM	1:00:00 AM	=MOD(B18,1)
19	01/02/2017 2:00:30 PM	2:00:30 PM	=MOD(B19,1)
20	03/03/2017 11:30:00 AM	11:30:00 AM	=MOD(B20,1)
21	04/01/2017 12:00:00 AM	12:00:00 AM	=MOD(B21,1)
22	05/10/2017 9:00:00 AM	9:00:00 AM	=MOD(B22,1)

\*Special thanks to Sunny Kow for reminding me about this tip.

## COMMON DATE CALCULATIONS


### CALCULATE AGE OR YEARS OF SERVICE

	B	C	D	E
15	Start Date	Years of Service	Calculation	Formula
16	1/03/2006	11	Whole Years	=YEAR(TODAY())-YEAR(B16)
17	1/03/2006	11	Whole Years (DATEDIF)	=DATEDIF(B17,TODAY(),"Y")
18	1/03/2006	11.62491444	Fractions of years	=YEARFRAC(B18,TODAY(),1)

**Warning:** The DATEDIF function can return errors.

	B	C	D
20	Birth Date	Age	Formula
21	1/03/2006	11 Years old	=YEAR(TODAY())-YEAR(B21)
22	1/03/2006	11 Years old	=DATEDIF(B22,TODAY(),"Y")
23	1/03/2006	11 years, 7 months, 19 days	=DATEDIF(B23,TODAY(),"Y")&" years, "&DATEDIF(B23,TODAY(),"YM")&" months, "&DATEDIF(B23,TODAY(),"MD")&" days"

**Tip:** use a custom number format **0 "Years old"** to add the text; 'Years old' to the age in column C.



**DIFFERENCE BETWEEN DATES**

	B	C	D	E	F
28	Start Date	End Date	Difference In	Formula Result	Formula
29	1/01/2017	4/10/2017	Days	276	=C29-B29
30	1/01/2017	4/10/2017	Days	276	=DAYS(C30,B30) *new function in Excel 2013
31					
32	31/12/2016	1/07/2017	Weeks	26	=(C32-B32)/7
33					
34	31/12/2016	1/07/2017	Whole Months Rounded Up	7	=(YEAR(C34)- YEAR(B34))*12+MONTH(C34)- MONTH(B34)
35	31/12/2016	1/07/2017	Whole Months Rounded Down	6	=IF(DAY(C35)>=DAY(B35),0,- 1)+(YEAR(C35)- YEAR(B35))*12+MONTH(C35)- MONTH(B35)
36	31/12/2016	1/07/2017	Whole Months (DATEDIF)	6	=DATEDIF(B36,C36,"M")
37					
38	1/01/2017	4/10/2020	Whole Years (DATEDIF)	3	=DATEDIF(B38,C38,"Y")
39	1/01/2017	4/10/2020	Whole Years Rounded Down	3	=YEAR(C39)-YEAR(B39)
40	1/01/2017	4/10/2020	Whole Years Rounded Up	4	=ROUNDUP( YEARFRAC(B40,C40,1),0)
41	1/01/2017	4/10/2020	Fractions of Years (YEARFRAC)	3.75633128	=YEARFRAC(B41,C41,1)



## FINDING DATES

	B	C	D	E
45	Date	Find	Formula Result	Formula
46	15/01/2017	Number of days in a month	31	=DAY(EOMONTH(B46,0))
47	15/01/2017	First day of month	1/01/2017	=EOMONTH(B47,-1)+1
48	15/01/2017	Last day of month	31/01/2017	=EOMONTH(B48,0)
49	15/01/2017	Last day of previous month	31/12/2016	=EOMONTH(B49,-1)
50	15/01/2017	First weekday of month	2/01/2017	=DATE(YEAR(B50),MONTH(B50),1)+CHOOSE(WEEKDAY(DATE(YEAR(B50),MONTH(B50),1)),1,0,0,0,0,2)
51	15/01/2017	Last weekday of month	31/01/2017	=DATE(YEAR(B51),MONTH(B51)+1,0)-(MAX(0,WEEKDAY(DATE(YEAR(B51),MONTH(B51)+1,0),2)-5))
52	20/02/2017	Calendar quarter of a date	1	=ROUNDUP(MONTH(B52)/3,0)
53	15/01/2017	Find last Thursday of Month	26/01/2017	=EOMONTH(B53,0)+1-WEEKDAY(EOMONTH(B53,0)+1-5)

**Tip:** 5 in the last example above, on row 53, represents Thursday. Change the value to find different days. E.g. Sunday = 1 and Saturday =7.

## DIFFERENCE BETWEEN TIMES (SAME DAY)

	B	C	D	E	F
57	Start Time	End Time	Difference In	Formula Result	Formula
58	9:00:00 AM	11:30:00 PM	Time, formatted hh:mm:ss	14:30:00	=C58-B58
59	9:00:00 AM	11:30:00 PM	Time, serial number	0.604166667	=C59-B59
60	9:00:00 AM	11:30:00 PM	Minutes, formatted [m]	870	=C60-B60
61	9:00:00 AM	11:30:00 PM	Seconds, formatted [ss]	52200	=C61-B61

### DIFFERENCE BETWEEN TIMES (DIFFERENT DAYS)

When time starts on the day before the end time, as is common in shift work, the start and end time either need to be entered with the date. As shown on row 66, or you require a different formula to account for the missing date information. As shown in cell E68.

	B	C	D	E	F
65	Start Time	End Time	Difference In	Formula Result	Formula
66	01/01/2017 10:00:00 PM	02/01/2017 9:00:00 AM	Time, formatted hh:mm:ss	11:00:00	=C66-B66
67					
68	10:00:00 PM	9:00:00 AM	Time, formatted hh:mm:ss	11:00:00	=IF(B68>C68,C68+1-B68,C68-B68)
69					
70	01/01/2017 10:00:00 PM	04/01/2017 9:30:45 AM	Days & Time, formatted d.hh:mm:ss	2.11:30:45	=C70-B70
71					
72	01/01/2017 10:00:00 PM	04/01/2017 9:30:45 AM	Days & Time, formatted d "days" hh "hours" mm "minutes"	2 days 11 hours 30 minutes	=C72-B72

### ROUNDING TIME AND TIME INTERVALS (USING CEILING & FLOOR)

	B	C	D	E	F
73	Start Time	End Time	Round to	Formula Result	Formula
74	9:00:00 AM	11:30:00 PM	Down to Previous Hour	14:00:00	=TIME(HOUR(C74-B74),FLOOR(MINUTE(C74-B74),60),0)
75	9:00:00 AM	11:30:00 PM	Up to Next Hour	15:00:00	=TIME(HOUR(C75-B75),CEILING(MINUTE(C75-B75),60),0)
76	9:00:00 AM	11:25:00 PM	Up to Next Half Hour	14:30:00	=TIME(HOUR(C76-B76),CEILING(MINUTE(C76-B76),30),0)
77	9:00:00 AM	11:05:00 PM	Up to next 6 minute interval	14:06:00	=TIME(HOUR(C77-B77),CEILING(MINUTE(C77-B77),6),0)
78	9:00:00 AM	11:05:00 PM	Down to previous 6 minute interval	14:00:00	=TIME(HOUR(C78-B78),FLOOR(MINUTE(C78-B78),6),0)

**Tip:** Alter the minutes entered in red to change the rounding intervals.

### ROUNDING TIME AND TIME INTERVALS (USING MROUND)

	B	C	D	E	F
82	Start Time	End Time	Round to	Formula Result	Formula
83	9:00:00 AM	11:30:00 PM	Down to Previous Hour	14:00:00	=TIME(HOUR(C83-B83),MROUND(MINUTE(C83-B83),0),0)
84	9:00:00 AM	11:25:00 PM	To <b>closest</b> Hour	14:00:00	=TIME(HOUR(C84-B84),MROUND(MINUTE(C84-B84),60),0)
85	9:00:00 AM	11:25:00 PM	To <b>closest</b> Half Hour	14:30:00	=TIME(HOUR(C85-B85),MROUND(MINUTE(C85-B85),30),0)
86	9:00:00 AM	11:05:00 PM	To <b>closest</b> 6 minute interval	14:06:00	=TIME(HOUR(C86-B86),MROUND(MINUTE(C86-B86),6),0)

**Tip:** Alter the minutes entered in red to change the rounding intervals.

### CONVERT TIME TO DECIMALS

Often, we want to use time in other calculations that require a decimal format, like wage calculations. For this we need to convert time to its decimal equivalent. Since we know there are 24 hours in a day, and 60 minutes in an hour and 60 seconds in a minute, we can multiply the time serial number to convert it to a decimal equivalent. The table below provides the logic for converting to hours, minutes or seconds.

	B	C	D	E	F
90	Time hh:mm:ss	Convert to	Multiply by	Formula Result	Formula
91	22:30:00	Hours	24	22.5	=B91*D91
92	22:30:00	Minutes	1440	1350	=B92*D92
93	22:30:00	Seconds	86400	81000	=B93*D93

### CONVERT MINUTES TO HOURS, HOURS TO DAYS ETC. IN DECIMALS

	B	C	D	E	F
97	Value	Unit	Convert to	Formula Result	Formula
98	12	Hours	Days	0.5	=B98/24
99	90	Minutes	Hours	1.5	=B99/60
100	180	Seconds	Minutes	3	=B100/60

## EXCEL DATE AND TIME FUNCTIONS

Date & Time Functions	Available From*	Syntax - FUNCTION(argument1, argument2, [argument3]...)	Description
<u>DATE</u>	Excel 2003	DATE(year,month,day)	Returns a date in serial number from separate year, month and day values
<u>TIME</u>	Excel 2003	TIME(hour,minute,second)	Returns a time serial number, from separate hour, minute and second values
<u>DATEVALUE</u>	Excel 2003	DATEVALUE(date_text)	Converts a date text string to a date serial number
<u>TIMEVALUE</u>	Excel 2003	TIMEVALUE(time_text)	Converts a time text string to a time serial number
<u>NOW</u>	Excel 2003	NOW()	Returns the current date & time
<u>TODAY</u>	Excel 2003	TODAY()	Returns today's date
<u>HOUR</u>	Excel 2003	HOUR(serial_number)	Returns the hour part of a time serial number
<u>MINUTE</u>	Excel 2003	MINUTE(serial_number)	Returns the minute part of a time serial number
<u>SECOND</u>	Excel 2003	SECOND(serial_number)	Returns the seconds part of a time serial number
<u>DAY</u>	Excel 2003	DAY(serial_number)	Returns the day (of the month) from a date serial number

Date & Time Functions	Available From*	Syntax - FUNCTION(argument1, argument2,[argument3]...)	Description
<a href="#"><u>MONTH</u></a>	Excel 2003	MONTH(serial_number)	Returns the month from a date serial number
<a href="#"><u>YEAR</u></a>	Excel 2003	YEAR(serial_number)	Returns the year from a date serial number
<a href="#"><u>WEEKNUM</u></a>	Excel 2003	WEEKNUM(serial_number, [return_type])	Returns an integer representing the week number (from 1 to 54) of the year from a date serial number
<a href="#"><u>ISOWEEKNUM</u></a>	Excel 2013	ISOWEEKNUM(date)	Returns the ISO week number of the year for a given date
<a href="#"><u>WEEKDAY</u></a>	Excel 2003	WEEKDAY(serial_number, [return_type])	Returns an integer representing the day of the week from a date serial number
<a href="#"><u>EDATE</u></a>	Excel 2003	EDATE(start_date,months)	Returns a date that is the specified number of months before or after an initial supplied start date
<a href="#"><u>EOMONTH</u></a>	Excel 2003	EOMONTH(start_date,months)	Returns a date that is the last day of the month that is a specified number of months before or after the date serial number
<a href="#"><u>WORKDAY</u></a>	Excel 2003	WORKDAY(start_date,days, [holidays])	Returns a date a number of working days (excluding weekends & holidays) before or after a given start date
<a href="#"><u>WORKDAY.INTL</u></a>	Excel 2010	WORKDAY.INTL(start_date,days, [weekend] , [holidays])	Returns a date a number of working days (excluding weekends & holidays) before or after a given start date, using supplied parameters to specify weekend days
<a href="#"><u>DAYS</u></a>	Excel 2013	DAYS(end_date,start_date)	Calculates the number of days between 2 dates
<a href="#"><u>DAYS360</u></a>	Excel 2003	DAYS360(start_date, end_date, [method])	Calculates the number of days between 2 dates, based on a 360-day year (12 x 30-day months)

Date & Time Functions	Available From*	Syntax - FUNCTION(argument1, argument2,[argument3]...)	Description
<a href="#"><u>NETWORKDAYS</u></a>	Excel 2003	NETWORKDAYS(start_date, end_date, [holidays])	Returns the number of whole networkdays (excluding weekends & holidays), between two dates
<a href="#"><u>NETWORKDAYS.INTL</u></a>	Excel 2010	NETWORKDAYS.INTL(start_date,end_date, [weekend], [holidays])	Returns the number of whole networkdays (excluding weekends & holidays), between two dates, using parameters to specify weekend days
<a href="#"><u>YEARFRAC</u></a>	Excel 2003	YEARFRAC(start_date, end_date, [basis])	Calculates the fraction of the year represented by the number of whole days between two dates
<a href="#"><u>DATEDIF</u></a>	Excel 2003	DATEDIF(start_date,end_date,unit)	<p>Calculates the number of days, months, or years between two dates. This function is useful in formulas where you need to calculate an age.</p> <p><b>Warning:</b> Excel provides the DATEDIF function in order to support older workbooks from Lotus 1-2-3. The DATEDIF function may calculate incorrect results under certain scenarios. Please see the DATEDIF section for further details.</p>

## DATE FUNCTION

The DATE function combines separate year, month and day values and converts them into a date serial number formatted as a date.

<b>Syntax:</b>	=DATE(year, month, day)
<b>year</b>	The value of the <b>year</b> argument can include one to four digits from 1900 to 9999.
<b>month</b>	A positive or negative integer representing the month of the year from 1 to 12 (January to December).
<b>day</b>	A positive or negative integer representing the day of the month from 1 to 31.

	B	C	D	E	F	G
9	<b>Day</b>	<b>Month</b>	<b>Year</b>	<b>DATE Formula</b>	<b>DATE Formula Result</b>	<b>Date Serial Number</b>
10	15	3	2018	=DATE(D10,C10,B10)	15/03/2018	43174

**Tip:** We can use the DATE function to add or subtract months or years to a date. See below.

	B	C	D	E	F	G
14	<b>Day</b>	<b>Month</b>	<b>Year</b>	<b>Increment by</b>	<b>DATE Formula Result</b>	<b>Date Formula</b>
15	15	3	2018	+2 years	15/03/2020	=DATE(D15+2,C15,B15)
16	15	3	2018	+3 months	15/06/2018	=DATE(D16,C16+3,B16)
17	15	3	2018	-2 years	15/03/2016	=DATE(D17-2,C17,B17)
18	15	3	2018	-3 months	15/12/2017	=DATE(D18,C18-3,B18)

**Tip:** We can also use the EDATE function to roll dates forward by a set number of months.

**Note:** If we add days and they exceed the number of days in the month, DATE will add the excess days to the start of the following month. See examples below.

	B	C	D	E	F	G
23	<b>Day</b>	<b>Month</b>	<b>Year</b>	<b>Increment by</b>	<b>DATE Formula Result</b>	<b>Date Formula</b>
24	15	3	2018	+10 days	25/03/2018	=DATE(D24,C24,B24+10)
25	25	3	2018	+20 days	14/04/2018	=DATE(D25,C25,B25+20)
26	25	3	2018	+300 days	19/01/2019	=DATE(D26,C26,B26+300)



The Year argument must be a positive value, but month and day can be negative! If **month** is greater than 12, **month** adds that number of months to the first month in the year specified.

If **month** is less than 1, **month** subtracts the magnitude of that number of months, plus 1, from

the first month in the year specified. See examples below.

	B	C	D	E	F	G
30	<b>Day</b>	<b>Month</b>	<b>Year</b>	<b>DATE Formula</b>	<b>DATE Formula Result</b>	<b>Date Serial Number</b>
31	15	-5	2018	=DATE(D31,C31,B31)	15/07/2017	42931
32	-15	4	2018	=DATE(D32,C32,B32)	16/03/2018	43175
33	40	15	2018	=DATE(D33,C33,B33)	9/04/2019	43564





## TIME FUNCTION

The TIME function combines separate hour, minute and second values and converts them into a time serial number. If the cell format was General before the formula is entered, the result is formatted as a date.

<b>Syntax:</b>	=TIME(hour, minute, second)
<b>hour</b>	A number from 0 (zero) to 32767 representing the hour. Any value greater than 23 will be divided by 24 and the remainder will be treated as the hour value. For example, TIME(27,0,0) = TIME(3,0,0) = .125 or 3:00 AM.
<b>minute</b>	A number from 0 to 32767 representing the minute. Any value greater than 59 will be converted to hours and minutes. For example, TIME(0,750,0) = TIME(12,30,0) = .520833 or 12:30 PM.
<b>second</b>	A number from 0 to 32767 representing the second. Any value greater than 59 will be converted to hours, minutes, and seconds. For example, TIME(0,0,2000) = TIME(0,33,22) = .023148 or 12:33:20 AM

	B	C	D	E	F	G
8	<b>Second</b>	<b>Minute</b>	<b>Hour</b>	<b>TIME Formula</b>	<b>TIME Formula Result</b>	<b>TIME Serial Number</b>
9	15	30	10	=TIME(D9,C9,B9)	10:30:15 AM	0.437673611

**Tip:** We can use the TIME function to add or subtract hours, minutes or seconds to a time. See below.

	B	C	D	E	F	G
13	<b>Second</b>	<b>Minute</b>	<b>Hour</b>	<b>Increment by</b>	<b>TIME Formula Result</b>	<b>TIME Formula</b>
14	0	10	15	+2 hours	5:10:00 PM	=TIME(D14+2,C14,B14)
15	15	0	12	+30 minutes	12:30:15 PM	=TIME(D15,C15+30,B15)
16	30	45	1	+60 minutes	2:45:30 AM	=TIME(D16,C16+60,B16)
17	0	30	4	-15 minutes	4:15:00 AM	=TIME(D17,C17-15,B17)
18	0	0	2	-3 hours	#NUM!	=TIME(D18-3,C18,B18)

Notice the last result shows the #NUM! error. This is because it results in a negative time and Excel can't calculate that. More on how to solve this later.

**Note:** If we add time that takes us past 24 hours it will add the excess time to 0 time, effectively starting the clock again. See examples below.

	B	C	D	E	F	G
23	<b>Second</b>	<b>Minute</b>	<b>Hour</b>	<b>Increment by</b>	<b>TIME Formula Result</b>	<b>TIME Formula</b>
24	0	30	22	+3 hours	1:30 AM	=TIME(D24+3,C24,B24)
25	0	15	23	+50 minutes	12:05 AM	=TIME(D25,C25+50,B25)

## DATEVALUE FUNCTION

The DATEVALUE function converts dates stored as text to a date serial number.

*Remember; my dates are dd/mm/yyyy*

<b>Syntax:</b>	= DATEVALUE(date_text)
----------------	------------------------

	B	C	D	E
	<b>Date stored as Text</b>	<b>DATEVALUE</b>	<b>Date Serial Number</b>	<b>Formula in</b>
8	<b>dd/mm/yyyy</b>	<b>Formula</b>	<b>Returned</b>	<b>Column C</b>
9	15/3/2017	15/03/2017	42809	=DATEVALUE(B9)
10	15.03.2017	#VALUE!	#VALUE!	=DATEVALUE(B10)
11	15-3-2017	15/03/2017	42809	=DATEVALUE(B11)
12	2017/01/15	15/01/2017	42750	=DATEVALUE(B12)
13	15 May 2017	15/05/2017	42870	=DATEVALUE(B13)
14	15 May	15/05/2017	42870	=DATEVALUE(B14)

**Tip 1:** Notice DATEVALUE doesn't like dates separated with periods.

**Tip 2:** if you don't have the year present Excel will input the year based on your PC's built in clock!

So, you can see the DATEVALUE function can fix many date formats, but you may find it doesn't work with everything. Don't worry, I'll cover more ways to fix dates formatted as text soon.

## TIMEVALUE FUNCTION

The TIMEVALUE function converts times stored as text to a date-time serial number.

<b>Syntax:</b>	= TIMEVALUE(time_text)
----------------	------------------------

**Tip:** The TIMEVALUE function is handy for extracting the time portion from a date-time text value.

	B	C	D	E
8	<b>Time stored as Text</b>	<b>TIMEVALUE Formula</b>	<b>Time Serial Number</b>	<b>Formula in Column C</b>
9	8:50 PM	8:50:00 PM	0.868055556	=TIMEVALUE(B9)
10	20:50	8:50:00 PM	0.868055556	=TIMEVALUE(B10)
11	15-3-2017 8:50 PM	8:50:00 PM	0.868055556	=TIMEVALUE(B11)
12	8:50:30 PM	8:50:30 PM	0.868402778	=TIMEVALUE(B12)
13	8:30PM	#VALUE!	#VALUE!	=TIMEVALUE(B13)
14	8:30pm	#VALUE!	#VALUE!	=TIMEVALUE(B14)

**Note:** TIMEVALUE can't fix times where the AM/PM isn't entered with a space after the time.

## NOW FUNCTION

The NOW function is unusual in that it doesn't have any arguments. It simply returns the current date and time from your computer clock.

**Caution:** The NOW function is volatile, meaning it recalculates everytime virtually anything changes in your workbook...best use it sparingly!

<b>Syntax:</b>	=NOW()
----------------	--------

C11		=NOW()
	B	C
11	<b>Formula result:</b>	15/10/2017 20:09

**Tip:** When you enter the NOW function Excel will format the cell in the date - time format based on your regional settings, (where no prior formatting was applied to the cell).

Use NOW to calculate the difference between a date-time serial number and now. e.g. how many hours since x or what will the time be in x hours.

	B	C	D
	<b>h:mm Since Start</b>		
21	<b>Start Date &amp; Time</b>	<b>Date &amp; Time</b>	<b>Formula</b>
22	12/10/2017 13:30	78:39	=NOW()-B22


	B	C	D
24	<b>Date &amp; Time Now</b>	<b>The Time in 7 hours</b>	<b>Formula</b>
25	12/10/2017 1:30:18 PM	16/10/2017 3:09:36 AM	16/10/2017 3:09

## TODAY FUNCTION

The TODAY function is unusual in that it doesn't have any arguments. It simply returns the current date serial number from your computer clock.

**Caution:** The TODAY function is volatile, meaning it recalculates every time virtually anything changes in your workbook...best use it sparingly!

<b>Syntax:</b>	=TODAY()
----------------	----------



C10	=TODAY()
B	C
10	Formula result: 15/10/2017

**Tip:** When you enter the TODAY function Excel will format the cell in the date format based on your regional settings, (where no prior formatting was applied to the cell).

Use TODAY to calculate the difference between a date serial number and today's date. e.g. how many days since x, or what will the date be in x days' time.

	B	C	D
19	<b>Start Date</b>	<b>Days Since Start Date</b>	<b>Formula</b>
20	1/03/2006	4246	=TODAY()-B20

	B	C	D
22	<b>Date Today</b>	<b>Date in 10 days' time</b>	<b>Formula</b>
23	15/10/2017	25/10/2017	=TODAY()+10

## HOUR FUNCTION

The HOUR function returns the hour integer from a time serial number ranging from 0 to 24.

<b>Syntax:</b>	=HOUR(serial_number)
----------------	----------------------

	B	C	D
8	<b>Time Serial Number</b>	<b>HOUR Formula Result</b>	<b>HOUR Formula</b>
9	5:10:00 PM	17	=HOUR(B9)
10	12:30:15 PM	12	=HOUR(B10)
11	0.5	12	=HOUR(B13)
12	4:15:00 AM	4	=HOUR(B12)
13	1/01/2018	0	=HOUR(B13)

**Tip:** cell C11 returns 12 because 12 hours is 50% of 24 hours.

The HOUR function will also accept a text string, no need to convert the text to a date-time serial number first e.g. "4:15 AM"

## MINUTE FUNCTION

The MINUTE function returns the minute integer, ranging from 0 to 59, from a time serial number.

<b>Syntax:</b>	=MINUTE(serial_number)
----------------	------------------------

	B	C	D
8	<b>Time Serial Number</b>	<b>MINUTE Formula Result</b>	<b>MINUTE Formula</b>
9	5:10:00 PM	10	=MINUTE(B9)
10	12:30:15 PM	30	=MINUTE(B10)
11	2:45:30 AM	45	=MINUTE(B11)
12	4:15:00 AM	15	=MINUTE(B12)
13	1/01/2018	0	=MINUTE(B13)

**Tip:** 0 is returned in cell C13 because there is no time portion to this date.

The MINUTE function will also accept a text string, no need to convert the text to a date-time serial number first e.g. "4:15 AM"

## SECOND FUNCTION

The SECOND function returns the second integer from a time serial number ranging from 0 to 59.

<b>Syntax:</b>	=SECOND(serial_number)
----------------	------------------------

	B	C	D
8	<b>Time Serial Number</b>	<b>SECOND Formula Result</b>	<b>SECOND Formula</b>
9	5:10:00 PM	0	=SECOND(B9)
10	12:30:15 PM	15	=SECOND(B10)
11	2:45:30 AM	30	=SECOND(B11)
12	4:15:58 AM	58	=SECOND(B12)
13	1/01/2018	0	=SECOND(B13)

**Tip:** 0 is returned in cell C13 because there is no time portion to this date.

The SECOND function will also accept a text string, no need to convert the text to a date-time serial number first e.g. "4:15:58 AM"



## DAY FUNCTION

The DAY function returns the day of the month between 1 and 31 from a date serial number or text.

<b>Syntax:</b>	=DAY(serial_number)
----------------	---------------------

	B	C	D
8	<b>Date</b>	<b>DAY Formula Result</b>	<b>DAY Formula</b>
9	15/03/2017	15	=DAY(B9)
10	25/12/2018	25	=DAY(B10)
11	15-3-2017	15	=DAY(B11)
12	29 Feb 2020	29	=DAY(B12)
13	15 May 2017	15	=DAY(B13)
14	15 May	15	=DAY(B14)

**Tip:** Rows 11:14 contain dates as text in column B, but DAY is still able to return the day! Don't rely on this though as results may be hit and miss.

You can also nest the DATE Function inside DAY, like so:

Formula: =DAY( DATE(2017,03,15) )

Result: =15

Or DATEVALUE like so:

Formula: =DAY( DATEVALUE("2017/03/15") )

Result: =15

## MONTH FUNCTION

The MONTH function returns the month of a date serial number between 1 and 12, representing January through December.

<b>Syntax:</b>	=MONTH(serial_number)
----------------	-----------------------

	B	C	D
8	<b>Date</b>	<b>MONTH Formula Result</b>	<b>MONTH Formula</b>
9	15/03/2017	3	=MONTH(B9)
10	25/12/2018	12	=MONTH(B10)
11	15-3-2017	3	=MONTH(B11)
12	29 Feb 2020	2	=MONTH(B12)
13	10 May 2017	5	=MONTH(B13)
14	25 June	6	=MONTH(B14)
15	25 Jun	6	=MONTH(B15)

**Tip:** Rows 11:15 contain dates as text in column B, but MONTH is still able to return the month number! We can exploit this feature to convert text months to month numbers which I'll show you later.

You can also nest the DATE Function inside MONTH, like so:

Formula: =MONTH( DATE(2017,03,15) )

Result: =3

Or DATEVALUE like so:

Formula: =MONTH( DATEVALUE("2017/03/15") )

Result: =3

## YEAR FUNCTION

The YEAR function returns the year from a date serial number as an integer between 1900 and 9999.

<b>Syntax:</b>	=YEAR(serial_number)
----------------	----------------------

	B	C	D
8	<b>Date</b>	<b>YEAR Formula Result</b>	<b>YEAR Formula</b>
9	15/03/2017	2017	=YEAR(B9)
10	25/12/2018	2018	=YEAR(B10)
11	15-3-2017	2017	=YEAR(B11)
12	29 Feb 2020	2020	=YEAR(B12)
13	10 May 2017	2017	=YEAR(B13)
14	25 June	2017	=YEAR(B14)
15	25 Jun	2017	=YEAR(B15)

**Tip:** Rows 11:15 contain dates as text in column B, but YEAR is still able to return the year number! Where a year isn't specified Excel will assume the year from your PC clock. Don't rely on dates stored as text as results may be hit and miss.

You can also nest the DATE Function inside YEAR, like so:

Formula: =YEAR( DATE(2017,03,15) )

Result: =2017

Or DATEVALUE like so:

Formula: =YEAR( DATEVALUE("2017/03/15") )

Result: =2017

## WEEKNUM FUNCTION

The WEEKNUM function returns the week number (between 1 and 54) of a date serial number. e.g. 1st Jan 2017 is in week number 1 of the year.


There are two types of systems available with this function:

- **System 1** where the week containing January 1 is the first week of the year, and is numbered week 1.
- **System 2** starts with the first Thursday of the year being in week 1. This system is the methodology specified in ISO 8601, which is commonly known as the European week numbering system.

<b>Syntax:</b>	= WEEKNUM(serial_number, [return_type])
----------------	---

**Note:** The return type argument is optional, if omitted it will default to return type 1.

Below is a list of the return type options available:



**Tip:** Return types 2 and 11 for Monday are the same as one another. Type 2 is only included for backward compatibility with earlier versions of Excel. Likewise, return type 1 for Sunday.

The next page contains some examples of WEEKNUM formulas using different return types.

			WEEKNUM Formula
Date	Return Type	Return Type Day	Result
Sun, Jan 01, 2017	1	Sunday	1
Sun, Jan 01, 2017	2	Monday	1
Sun, Jan 01, 2017	11	Monday	1
Sun, Jan 01, 2017	12	Tuesday	1
Sun, Jan 01, 2017	13	Wednesday	1
Sun, Jan 01, 2017	14	Thursday	1
Sun, Jan 01, 2017	15	Friday	1
Sun, Jan 01, 2017	16	Saturday	1
Sun, Jan 01, 2017	17	Sunday	1
Sun, Jan 01, 2017	21	Monday	52
Mon, Mar 13, 2017	1	Sunday	11
Mon, Mar 13, 2017	2	Monday	12
Mon, Mar 13, 2017	11	Monday	12
Mon, Mar 13, 2017	12	Tuesday	11
Mon, Mar 13, 2017	13	Wednesday	11
Mon, Mar 13, 2017	14	Thursday	11
Mon, Mar 13, 2017	15	Friday	11
Mon, Mar 13, 2017	16	Saturday	11
Mon, Mar 13, 2017	17	Sunday	11
Mon, Mar 13, 2017	21	Monday	11
Thu, Jan 05, 2017	1	Sunday	1
Thu, Jan 05, 2017	2	Monday	2
Thu, Jan 05, 2017	11	Monday	2
Thu, Jan 05, 2017	12	Tuesday	2
Thu, Jan 05, 2017	13	Wednesday	2
Thu, Jan 05, 2017	14	Thursday	2
Thu, Jan 05, 2017	15	Friday	1
Thu, Jan 05, 2017	16	Saturday	1
Thu, Jan 05, 2017	17	Sunday	1
Thu, Jan 05, 2017	21	Monday	1

## ISOWEEKNUM FUNCTION

The ISOWEEKNUM function (new in Excel 2013) returns the week number of a date serial number. The first Monday of the year marks the start of week 1.

**Tip:** it also works with dates input as Text, but this could be unreliable so it's always best to work with date serial numbers.

<b>Syntax:</b>	=ISOWEEKNUM(date)
----------------	-------------------

	B	C	D
11	<b>Date</b>	<b>ISOWEEKNUM Formula Result</b>	<b>Formula</b>
12	Sun, Jan 01, 2017	52	=ISOWEEKNUM(B12)
13	Mon, Jan 02, 2017	1	=ISOWEEKNUM(B13)
14	Tue, Jan 03, 2017	1	=ISOWEEKNUM(B14)
15	Wed, Jan 04, 2017	1	=ISOWEEKNUM(B15)
16	Thu, Jan 05, 2017	1	=ISOWEEKNUM(B16)
17	Fri, Jan 06, 2017	1	=ISOWEEKNUM(B17)
18	Sat, Jan 07, 2017	1	=ISOWEEKNUM(B18)
19	Sun, Jan 08, 2017	1	=ISOWEEKNUM(B19)
20	Mon, Jan 09, 2017	2	=ISOWEEKNUM(B20)
21	Tue, Jan 10, 2017	2	=ISOWEEKNUM(B21)
22	Wed, Jan 11, 2017	2	=ISOWEEKNUM(B22)
23	Thu, Jan 12, 2017	2	=ISOWEEKNUM(B23)

## WEEKDAY FUNCTION

The WEEKDAY function returns the day number of the week from a date serial number. The default return type ranges from 1 for Sunday, through to 7 for Saturday, but you can choose from a list of return types.

<b>Syntax:</b>	=WEEKDAY(serial_number, [return_type])
<b>serial_number</b>	A date serial number
<b>return_type</b>	<p>The return type argument is optional, if omitted it will default to return type 1.</p> <p><code>=WEEKDAY(DATE(2017,03,15),</code>  <code>WEEKDAY(serial_number, [return_type])</code></p> <ul style="list-style-type: none"> <li>1 - Numbers 1 (Sunday) through 7 (Saturday)</li> <li>2 - Numbers 1 (Monday) through 7 (Sunday)</li> <li>3 - Numbers 0 (Monday) through 6 (Sunday)</li> <li>11 - Numbers 1 (Monday) through 7 (Sunday)</li> <li>12 - Numbers 1 (Tuesday) through 7 (Monday)</li> <li>13 - Numbers 1 (Wednesday) through 7 (Tuesday)</li> <li>14 - Numbers 1 (Thursday) through 7 (Wednesday)</li> <li>15 - Numbers 1 (Friday) through 7 (Thursday)</li> <li>16 - Numbers 1 (Saturday) through 7 (Friday)</li> <li>17 - Numbers 1 (Sunday) through 7 (Saturday)</li> </ul>

**Tip:** Return types 1 and 17 for Sunday are the same as one another. Type 1 is only included for backward compatibility with earlier versions of Excel. Likewise, return type 2 and 3 for Monday.

	B	C	D	E	F
	Date	Return Type	Start - End	WEEKDAY Formula Result	Formula
23					
24	Sun, Jan 01, 2017	1	1 Sunday - 7 Saturday	1	=WEEKDAY(B24,C24)
25	Sun, Jan 01, 2017	2	1 Monday - 7 Sunday	7	=WEEKDAY(B25,C25)
26	Sun, Jan 01, 2017	3	0 Monday - 6 Sunday	6	=WEEKDAY(B26,C26)
27	Sun, Jan 01, 2017	11	1 Monday - 7 Sunday	7	=WEEKDAY(B27,C27)
28	Sun, Jan 01, 2017	12	1 Tuesday - 7 Monday	6	=WEEKDAY(B28,C28)
29	Sun, Jan 01, 2017	13	1 Wednesday - 7 Tuesday	5	=WEEKDAY(B29,C29)
30	Sun, Jan 01, 2017	14	1 Thursday - 7 Wednesday	4	=WEEKDAY(B30,C30)
31	Sun, Jan 01, 2017	15	1 Friday - 7 Thursday	3	=WEEKDAY(B31,C31)
32	Sun, Jan 01, 2017	16	1 Saturday - 7 Friday	2	=WEEKDAY(B32,C32)
33	Sun, Jan 01, 2017	17	1 Sunday - 7 Saturday	1	=WEEKDAY(B33,C33)

## EDATE FUNCTION

The EDATE function rolls a date serial number (ignoring any time element) forward or back based on the number of months specified in the 'months' argument.

<b>Syntax:</b>	=EDATE( start_date, months)
<b>start_date</b>	A date serial number or text
<b>months</b>	The number of months before or after start_date. A positive value for months yields a future date; a negative value yields a past date.

	B	C	D	E
10	<b>Date</b>	<b>Months to +/-</b>	<b>EDATE Formula Result</b>	<b>Days Adjusted By</b>
11	15/03/2017	1	15/04/2017	31
12	15/03/2017	-1	15/02/2017	-28
13	29/02/2020	1	29/03/2020	29
14	29/02/2020	-1	29/01/2020	-31
15	1-1-17	12	1/01/2018	365
16	30/03/2017	-1	28/02/2017	-30
17	30/03/2017	-12	30/03/2016	-365
18	30/03/2017	6.8	30/09/2017	184



Notice the number of days varies depending on the month being adjusted.

**Tip:** EDATE works with dates stored as text (B15), but this is not reliable. It's always best to work with date serial numbers.

**Note:** only whole numbers are recognised by the 'months' argument. 6.8 in cell C16 is rounded down to 6.



## EOMONTH FUNCTION

The EOMONTH function returns the last day of the month before or after a start date specified with a date serial number (ignoring any time element).

<b>Syntax:</b>	=EOMONTH( start_date, months)
<b>start_date</b>	A date serial number or text
<b>months</b>	The number of months before or after start_date. A positive value for months yields a future date; a negative value yields a past date.

	B	C	D
			<b>EOMONTH Formula</b>
8	<b>Date</b>	<b>Months to +/-</b>	<b>Result</b>
9	15/03/2017	1	30/04/2017
10	15/03/2017	-1	28/02/2017
11	30/03/2017	0	31/03/2017
12	29/02/2020	1	31/03/2020
13	29/02/2020	-1	31/01/2020
14	1-1-17	12	31/01/2018
15	30/03/2017	-1	28/02/2017
16	30/03/2017	-12	31/03/2016

**Note:** only whole numbers are recognised by the 'months' argument. e.g. 1.7 months would be rounded down to 1.

**Tip:** EOMONTH works with dates stored as text (e.g. cell B14), but this is not reliable. It's always best to work with date serial numbers.

**Bonus Tip:** There's no 'STARTOFMONTH' function, but you can also use EOMONTH to find the start of a month, e.g. let's say your 'start date' is 30th January and you want to return 1st January. You simply use EOMONTH to find the end of the previous month and then add 1:

Formula: =EOMONTH(DATE(2017,01,30), -1) +1

Result: = 1st January 2017

## WORKDAY FUNCTION

The WORKDAY function returns a date serial number that is the specified number of working days before or after the start date. Working days exclude weekends and any dates identified as holidays.

WORKDAY is handy when calculating due dates that exclude weekends (Saturday & Sunday) or holidays.

<b>Syntax:</b>	=WORKDAY( start_date, days, [holidays])
<b>start_date</b>	A date serial number or text
<b>days</b>	Number of non-weekend and non-holidays to add or subtract
<b>[holidays]</b>	Optional list of holidays to exclude

	B	C	D	E	F	G
			=WORKDAY(B16,C16,\$G\$16:\$G\$23)			
			<b>WORKDAY Formula</b>			<b>Holiday</b>
15	<b>Start Date</b>	<b>Days to +/-</b>	<b>Result</b>			<b>Dates</b>
16	Thu, Dec 29, 2016	5	Fri, Jan 06, 2017			1/01/2017
17	Sun, Jan 01, 2017	5	Mon, Jan 09, 2017			2/01/2017
18	Thu, Mar 30, 2017	-5	Thu, Mar 23, 2017			26/01/2017
19	Wed, Apr 12, 2017	10	Mon, May 01, 2017			14/04/2017
20	Wed, Dec 27, 2017	-5	Mon, Dec 18, 2017			17/04/2017
21						25/04/2017
22						25/12/2017
23						26/12/2017


**Note 1:** only the date portion of a date-time serial number is used by WORKDAY. Any time element is ignored.

**Note 2:** only whole numbers are recognised by the 'days' argument. e.g. 1.7 days would be rounded down to 1 day.

## WORKDAY.INTL FUNCTION

The WORKDAY.INTL function (new in Excel 2010) returns a date serial number that is the specified number of working days before or after the start date. Working days exclude weekends, which can be customized, and any dates identified as holidays.

WORKDAY.INTL is handy when calculating due dates that exclude weekends or holidays where your weekend days aren't Saturday and Sunday.

<b>Syntax:</b>	=WORKDAY.INTL(start_date, days, [weekend], [holidays])
<b>start_date</b>	A date serial number or text
<b>days</b>	Number of non-weekend and non-holidays to add or subtract
<b>[weekend]</b>	Optional - choose which days of the week are not working days (see list below). If omitted, the default is Saturday and Sunday.  
<b>[holidays]</b>	Optional list of holidays to exclude

	B	C	D	E	F	G
	Start Date	Days to +/-	WORKDAY.INTL Formula Result			Holiday Dates
29						
30	Thu, Dec 29, 2016	5	Thu, Jan 05, 2017			1/01/2017
31	Sun, Jan 01, 2017	5	Sat, Jan 07, 2017			2/01/2017
32	Thu, Mar 30, 2017	-5	Fri, Mar 24, 2017			26/01/2017
33	Wed, Apr 12, 2017	10	Thu, Apr 27, 2017			14/04/2017
34	Wed, Dec 27, 2017	-5	Tue, Dec 19, 2017			17/04/2017
35						25/04/2017
36						25/12/2017
37						26/12/2017

**Note 1:** only the date portion of a date-time serial number is used by WORKDAY.INTL. Any time element is ignored.

**Note 2:** only whole numbers are recognised by the 'days' argument. e.g. 1.7 days would be rounded down to 1 day.

**Bonus Tip:** the weekend argument can be stipulated using a text string of 7 characters, with each character representing a day, starting with Monday. A 1 represents a non-working day and a 0 represents a workday. See examples below which stipulate non-workdays for Thursday, Friday, Saturday and Sunday.

The text string can be placed in a cell e.g. D50 or directly in the formula, see cell E51.

	B	C	D	E	F
	Start Date	Days to +/-	Weekend Days	WORKDAY.INTL Formula Result	Formula
49	1/01/2017	-5	0001111	20/12/2016	=WORKDAY.INTL(B50,C50,D50,\$G\$30:\$G\$37)
51	1/01/2017	-5		20/12/2016	=WORKDAY.INTL(B51,C51,"0001111",\$G\$30:\$G\$37)

## DAYS FUNCTION

The DAYS function (new in Excel 2013) returns the number of whole days between two date serial numbers. Ignores time portions of date-time serial numbers.

<b>Syntax:</b>	=DAYS( end_date, start_date)
<b>start_date</b>	A date serial number or text
<b>end_date</b>	A date serial number or text

	B	C	D	E
8	<b>Start Date</b>	<b>End Date</b>	<b>DAYS Formula Result</b>	<b>DAYS Formula</b>
9	1/01/2017	31/01/2017	30	=DAYS(C9,B9)
10	1/01/2017	31/12/2017	364	=DAYS(C10,B10)
11	1/01/2017	31/12/2017	364	=DAYS(C11,B11)
12	29/02/2020	29/01/2020	-31	=DAYS(C12,B12)
13	1-1-2017	31/01/2017	30	=DAYS(C13,B13)
14	30/03/2017	28/02/2017	-30	=DAYS(C14,B14)
15	30/03/2017	30/03/2016	-365	=DAYS(C15,B15)
16	30/03/2017	30/09/2017	184	=DAYS(C16,B16)



Notice the first day isn't included in the count. You need to add 1 to the result if you want inclusive dates.

**Tip:** DAYS works with dates stored as text (cell B13), but this is not reliable. It's always best to work with date serial numbers.

## DAYS360 FUNCTION

The DAYS360 function returns the number of days between two date serial numbers based on a 360-day year, or twelve 30-day months. This method is used in some accounting calculations.

<b>Syntax:</b>	=DAYS360(start_date, end_date, [method])
<b>start_date</b>	A date serial number or text
<b>end_date</b>	A date serial number or text
<b>[method]</b>	Optional logical value TRUE/FALSE – see more below.

The optional '**method**' argument is a logical value (TRUE/FALSE) that specifies whether to use the U.S. or European method. If omitted it is treated as FALSE.

**FALSE:** U.S. (NASD) method. If the starting date is the last day of a month, it becomes equal to the 30th day of the same month. If the ending date is the last day of a month and the starting date is earlier than the 30th day of a month, the ending date becomes equal to the 1st day of the next month; otherwise the ending date becomes equal to the 30th day of the same month.

**TRUE:** European method. Starting dates and ending dates that occur on the 31st day of a month become equal to the 30th day of the same month.

	B	C	D	E
	<b>DAYS360</b>			
22	<b>Start Date</b>	<b>End Date</b>	<b>Formula Result</b>	<b>DAYS360 Formula</b>
23	1/01/2017	31/01/2017	29	=DAYS360(B23,C23,TRUE)
24	1/01/2017	31/12/2017	359	=DAYS360(B24,C24,TRUE)
25	30/01/2017	1/02/2017	1	=DAYS360(B25,C25,TRUE)
26	29/02/2020	29/01/2020	-30	=DAYS360(B26,C26,TRUE)
27	1-1-2017	31/01/2017	29	=DAYS360(B27,C27,TRUE)
28	30/03/2017	28/02/2017	-32	=DAYS360(B28,C28,TRUE)
29	30/03/2017	30/03/2016	-360	=DAYS360(B29,C29,TRUE)
30				
31	1/01/2017	31/01/2017	30	=DAYS360(B31,C31,FALSE)
32	1/01/2017	31/12/2017	360	=DAYS360(B32,C32,FALSE)
33	30/01/2017	1/02/2017	1	=DAYS360(B33,C33,FALSE)
34	29/02/2020	29/01/2020	-31	=DAYS360(B34,C34,FALSE)
35	1-1-2017	31/01/2017	30	=DAYS360(B35,C35,FALSE)
36	30/03/2017	28/02/2017	-32	=DAYS360(B36,C36,FALSE)
37	30/03/2017	30/03/2016	-360	=DAYS360(B37,C37,FALSE)



Notice the difference between the TRUE/FALSE method results.

**Tip:** DAYS360 works with dates stored as text (cells B27 & B35), but this is not reliable. It's always best to work with date serial numbers.

**Note:** only the date portion of a date-time serial number is used by DAYS360. Any time element is ignored. As a result, only whole days are returned.

## NETWORKDAYS FUNCTION

The NETWORKDAYS function returns the number of whole working days between two date serial numbers, excluding weekends (Saturday & Sunday) and holidays.

<b>Syntax:</b>	=NETWORKDAYS(start_date, end_date, [holidays])
<b>start_date</b>	A date serial number or text
<b>end_date</b>	A date serial number or text
<b>[holidays]</b>	This optional argument can be a range of cells containing holiday dates that you want to exclude from the workday count.

In the example below, my list of holidays are in cells J11:J18.

	B	C	D	E
	<b>NETWORKDAYS</b>			
10	<b>Start Date</b>	<b>End Date</b>	<b>Formula Result</b>	<b>NETWORKDAYS Formula</b>
11	1/01/2017	31/01/2017	20	=NETWORKDAYS(B11,C11,\$J\$11:\$J\$18)
12	1/01/2017	31/12/2017	253	=NETWORKDAYS(B12,C12,\$J\$11:\$J\$18)
13	1/01/2017	4/01/2017	2	=NETWORKDAYS(B13,C13,\$J\$11:\$J\$18)
14	1/01/2017	2/01/2017	0	=NETWORKDAYS(B14,C14,\$J\$11:\$J\$18)
15	12/04/2017	20/04/2017	5	=NETWORKDAYS(B15,C15,\$J\$11:\$J\$18)
16	3/01/2017	4/01/2017	2	=NETWORKDAYS(B16,C16,\$J\$11:\$J\$18)
17	1-1-2017	31/01/2017	20	=NETWORKDAYS(B17,C17,\$J\$11:\$J\$18)

**Note:** NETWORKDAYS results are inclusive of the start and end dates.

**Tip:** NETWORKDAYS works with dates stored as text, but this is not reliable. It's always best to work with date serial numbers.

## NETWORKDAYS.INTL FUNCTION

The NETWORKDAYS.INTL function (new in Excel 2010) returns the number of working days between two date serial numbers, excluding weekends and holidays. Weekend days can be specified.

<b>Syntax:</b>	=NETWORKDAYS.INTL(start_date, end_date, [weekend], [holidays])
<b>[weekend]</b>	<p>This optional argument allows you to specify which days are included in the weekend. See list below:</p> <div style="border: 1px solid gray; padding: 5px; margin: 10px 0;"> <p>NETWORKDAYS.INTL(start_date, end_date, [weekend], [holidays])</p> <ul style="list-style-type: none"> <li>1 - Saturday, Sunday</li> <li>2 - Sunday, Monday</li> <li>3 - Monday, Tuesday</li> <li>4 - Tuesday, Wednesday</li> <li>5 - Wednesday, Thursday</li> <li>6 - Thursday, Friday</li> <li>7 - Friday, Saturday</li> <li>11 - Sunday only</li> <li>12 - Monday only</li> <li>13 - Tuesday only</li> <li>14 - Wednesday only</li> <li>15 - Thursday only</li> <li>16 - Friday only</li> <li>17 - Saturday only</li> </ul> </div>
<b>[holidays]</b>	This optional argument can be a range of cells containing holiday dates that you want to exclude from the workday count. In the example table below, my holidays are in cells J24:J31.

	B	C	D	E
				<b>NETWORKDAYS</b>
				<b>Formula Result</b>
23	<b>Start Date</b>	<b>End Date</b>	<b>Weekend Type</b>	
24	1/01/2017	31/01/2017	1	20
25	1/01/2017	31/01/2017	2	20
26	1/01/2017	31/01/2017	3	19
27	1/01/2017	31/01/2017	4	19
28	1/01/2017	31/01/2017	5	21
29	1/01/2017	31/01/2017	6	21
30	1/01/2017	31/01/2017	7	20
31	1/01/2017	31/01/2017	11	24
32	1/01/2017	31/01/2017	12	24
33	1/01/2017	31/01/2017	13	23
34	1/01/2017	31/01/2017	14	24
35	1/01/2017	31/01/2017	15	25
36	1/01/2017	31/01/2017	16	24
37	1/01/2017	31/01/2017	17	24



**Note:** NETWORKDAYS.INTL results are inclusive of the start and end dates.

**Tip:** NETWORKDAYS.INTL works with dates stored as text, but this is not reliable. It's always best to work with date serial numbers.

**Bonus Tip:** the weekend argument can be stipulated using a text string of 7 characters, with each character representing a day, starting with Monday. A 1 represents a non-working day and a 0 represents a workday. See examples below which stipulate non-workdays for Thursday, Friday, Saturday and Sunday.

The text string can be placed in a cell e.g. D45 or directly in the formula, see cell E46.

	B	C	D	E	F
	Start Date	End Date	Weekend Type	NETWORKDAYS Formula Result	Formula
44					
45	1/01/2017	31/01/2017	0001111	13	=NETWORKDAYS.INTL(B45,C45,D45,\$J\$24:\$J\$31)
46	1/01/2017	31/01/2017		13	=NETWORKDAYS.INTL(B46,C46,"0001111",\$J\$24:\$J\$31)

## YEARFRAC FUNCTION

The YEARFRAC function returns the year fraction representing the number of whole days between `start_date` and `end_date` serial numbers.

This function is useful for calculating the portion of a whole year for use in other calculations and is particularly useful in HR/employee related calculations.

<b>Syntax:</b>	=YEARFRAC(start_date, end_date, [basis])
<b>start_date</b>	A date serial number or text
<b>end_date</b>	A date serial number or text
<b>[basis]</b>	<p>This optional argument allows you to specify the day count method from the list of options below. If omitted, it defaults to 0.</p> <div style="border: 1px solid gray; padding: 5px; margin: 5px 0;"> <p>YEARFRAC(start_date, end_date, [basis])</p> <ul style="list-style-type: none"> <li>0 - US (NASD) 30/360</li> <li>1 - Actual/actual</li> <li>2 - Actual/360</li> <li>3 - Actual/365</li> <li>4 - European 30/360</li> </ul> </div>

	B	C	D	E
17	<b>Start Date</b>	<b>End Date</b>	<b>Basis</b>	<b>YEARFRAC Formula Result</b>
18	1/01/2017	31/03/2017	0	0.25
19	1/01/2017	31/03/2017	1	0.243835616
20	1/01/2017	31/03/2017	2	0.247222222
21	1/01/2017	31/03/2017	3	0.243835616
22	1/01/2017	31/03/2017	4	0.247222222
23	1/01/2017	31/03/2017		0.25
24	1/01/2017	31/03/2017	5	#NUM!
25	1/01/2017	31/03/2017	-1	#NUM!

**Tip:** You're most likely to use basis 1, unless your business uses a specific system.




**Note 1:** #NUM! Errors are returned if the basis argument is <0 or >4.

**Note 2:** only the date portion of a date-time serial number is used by YEARFRAC. Any time element is ignored. As a result, only whole days are returned.

## DATEDIF FUNCTION (USE WITH CAUTION)

For a long time, the DATEDIF function was undocumented. It is only included in Excel for backward compatibility with Lotus 1-2-3! If you try to enter the function it won't display in the auto-complete list, you must be brave and press on, confident in the fact that when you press ENTER it will evaluate!



DATEDIF calculates the number of days, months or years between two dates, but it is known to return incorrect results in some situations and is particularly buggy in Excel 2007. See details below.

**Syntax:** =DATEDIF(start\_date, end\_date, unit)

<b>Syntax:</b>	=DATEDIF(start_date, end_date, unit)
<b>start_date</b>	A date serial number
<b>end_date</b>	A date serial number
<b>unit</b>	The unit specifies what you want returned. The table below outlines the options.

Unit	Returns
"Y"	The number of <b>complete</b> years in the period.
"M"	The number of <b>complete</b> months in the period.
"D"	The number of days in the period.
"MD"	The difference between the days in start_date and end_date. The months and years of the dates are ignored.
"YM"	The difference between the months in start_date and end_date. The days and years of the dates are ignored.
"YD"	The difference between the days of start_date and end_date. The years of the dates are ignored.



**Caution!** The "M" and "MD" arguments have known limitations. See examples over.

**Note:** DATEDIF calculations do not include the start date in the resultant count, and they only return whole numbers.

**Tip:** when entering the Unit direct in the formula, surround it with double quotes e.g. =DATEDIF(start\_date, end\_date, "M")

	B	C	D	E
24	<b>Start Date</b>	<b>End Date</b>	<b>Unit</b>	<b>DATEDIF Formula Result</b>
25	1/01/2017	25/02/2020	Y	3
26	1/01/2017	25/02/2019	M	25
27	1/01/2017	3/01/2017	D	2
28	1/01/2017	5/02/2018	MD	4
29	1/01/2017	25/02/2019	YM	1
30	1/01/2017	25/02/2019	YD	55
31	<b>"M" Examples</b>			
32	1/01/2017	31/01/2017	M	0
33	1/01/2017	1/02/2017	M	1
34	1/01/2017	31/03/2017	M	2
35	1/01/2017	28/02/2017	M	1
36	31/12/2016	28/02/2017	M	1
37	31/12/2016	31/03/2017	M	3
38	<b>"MD" Examples</b>			
39	1/01/2017	31/01/2017	MD	30
40	1/01/2017	1/02/2017	MD	0
41	1/01/2017	28/02/2017	MD	27
42	1/01/2017	31/03/2017	MD	30
43	31/12/2016	1/03/2017	MD	-2
44	28/01/2017	1/02/2017	MD	4
45	28/01/2017	1/03/2017	MD	1



Cell E36 result should be 2! Something is not right with Feb 28, 2017.

Yikes! Cell E43 contains an example of MD returning erroneous results. It should be 1, not -2. Cells E44 & E45 also return inconsistent results.

## DATEDIF ALTERNATIVES

Now you see how unreliable DATEDIF can be, it's probably better to err on the side of caution and use a different function to achieve the same results. \*Note the M substitutes either round the months up or down. There is no direct equivalent to DATEDIF with Unit "M". Likewise, the alternate to MD doesn't replicate perfectly in every scenario.

	B	C	D	E	F
	DATEDIF			Alternate Formula	
50	Start Date	End Date	Unit	Result	Alternate Formula
51	1/01/2017	25/02/2020	Y	3	=YEAR(C51)-YEAR(B51)
52	31/01/2017	25/04/2017	M (round up*)	3	=(YEAR(C52)-YEAR(B52))*12+MONTH(C52)-MONTH(B52)
53	31/01/2017	25/04/2017	M (round down*)	2	=IF(DAY(C53)>=DAY(B53),0,-1)+(YEAR(C53)-
54	1/01/2017	3/01/2017	D	2	=C54-B54
55	31/01/2017	5/03/2017	MD	5	=IF(DAY(C55)<DAY(B55),DATE(YEAR(B55),MONTH(B55)+1,DAY(C55))-B55,DAY(C55)-DAY(B55))
56	1/01/2017	25/02/2019	YM	1	=MONTH(C56)-MONTH(B56)
57	1/01/2017	25/02/2019	YD	55	=DATE(YEAR(B57),MONTH(C57),DAY(C57))-B57

## MORE RESOURCES

### Weekly Newsletter

100 Tips & Tricks eBook & Newsletter..... <https://www.myonlinetraininghub.com/sign-up-for-100-excel-tips-and-tricks>

### Tutorials

Excel Formulas..... <https://www.myonlinetraininghub.com/excel-formulas>  
PivotTable Blog Posts..... <https://www.myonlinetraininghub.com/category/excel-pivottables>  
Power Query Blog Posts..... <https://www.myonlinetraininghub.com/category/power-query>  
Power Pivot Blog Posts..... <https://www.myonlinetraininghub.com/category/power-pivot>  
Excel Macros and VBA Blog Posts..... <https://www.myonlinetraininghub.com/category/excel-vba>  
Charting Blog Posts..... <http://www.myonlinetraininghub.com/category/excel-charts>  
Excel Dashboard Blog Posts..... <http://www.myonlinetraininghub.com/category/excel-dashboard>

### Free Webinars

Excel Dashboards..... <http://www.myonlinetraininghub.com/excel-webinars>  
Power BI..... [https://www.myonlinetraininghub.com/excel-dashboard-webinar-reg.htm#pbi\\_anchor](https://www.myonlinetraininghub.com/excel-dashboard-webinar-reg.htm#pbi_anchor)

### Support

Excel Forum..... <https://www.myonlinetraininghub.com/excel-forum>

### Courses

Complete Course Library..... <https://www.myonlinetraininghub.com/>  
Premium Bundle (Excel, Word & Outlook)..... <https://www.myonlinetraininghub.com/microsoft-office-online-training-courses>  
Advanced Excel..... <https://www.myonlinetraininghub.com/excel-expert-upgrade>  
PivotTables..... <https://www.myonlinetraininghub.com/excel-pivortable-course>

Power Query.....	<a href="https://www.myonlinetraininghub.com/excel-power-query-course">https://www.myonlinetraininghub.com/excel-power-query-course</a>
Power Pivot.....	<a href="https://www.myonlinetraininghub.com/power-pivot-course">https://www.myonlinetraininghub.com/power-pivot-course</a>
Excel for Decision Making.....	<a href="https://www.myonlinetraininghub.com/excel-for-decision-making-course">https://www.myonlinetraininghub.com/excel-for-decision-making-course</a>
Excel for Customer Service.....	<a href="https://www.myonlinetraininghub.com/excel-for-customer-service-professionals">https://www.myonlinetraininghub.com/excel-for-customer-service-professionals</a>
Excel for Finance.....	<a href="https://www.myonlinetraininghub.com/excel-for-finance-course">https://www.myonlinetraininghub.com/excel-for-finance-course</a>
Excel Analysis ToolPak.....	<a href="https://www.myonlinetraininghub.com/excel-analysis-toolpak-course">https://www.myonlinetraininghub.com/excel-analysis-toolpak-course</a>
Excel Dashboards.....	<a href="http://www.myonlinetraininghub.com/excel-dashboard-course">http://www.myonlinetraininghub.com/excel-dashboard-course</a>
Power BI.....	<a href="http://www.myonlinetraininghub.com/power-bi-course">http://www.myonlinetraininghub.com/power-bi-course</a>

## ABOUT



One of the most challenging aspects of Excel is working with times and dates. It's certainly one of the most common topics I get asked about.

So, I wanted to put together a comprehensive resource that you could use to learn and master working with dates and times, as well as a tool you could refer back to from time to time as required.

Please feel free to share this eBook with your friends and colleagues. The more people I can help, the better. And if you have any feedback or suggestions, I'd love to hear from you.

Mynda Treacy

P.S. If you're interested, [click here](#) to read more about my background.

© Copyright MyOnlineTrainingHub.com

## THANKS

I'd like to thank and give credit to the following sites that helped me understand date and time in Excel:

- Chip Pearson: <http://www.cpearson.com/excel/datetime.htm>
- Microsoft: <https://support.office.com/en-US/article/Date-and-time-functions-reference-fd1b5961-c1ae-4677-be58-074152f97b81>

And finally, I'd like to thank Sunny Kow for his tech checking and suggestions for additional topics.